VERSUCHE ZUR SIMULATION DES WINDES

FACHHOCHSCHULE HAMBURG
FACHBEREICH SEEFAHRT

Schiffsführungs- und Simulationsanlage SUSAN

Zellulares Netz: Windraster

W.-W. Scheuermann

5/85

```
1000  REM ----------------ZELLULARES NETZ: WINDFELD-
    ----------------------------
1001  HOME
1002  PRINT "1.INITIALSTART"
1003  PRINT "2.RESTART AUS DRUCKDATEI"
1004  PRINT
1005  INPUT "DEINE WAHL? ";W
1006 A1 = 11:A2 = 17:H = 1:N = 5
1007 DT = 1
1008 SK = 15
1009 T =  - 17
1010  DIM X(30,30): REM DRUECKE
1020  DIM XNEU(30,30)
1030  HCOLOR= 3
1040  ON W GOTO 1100,5100
1090  REM ----------------------------------------
    ----------------------------
1100 T = T + 1: FOR J = 1 TO 20
1101 X(0,J) = X(1,J): REM WAENDE
1102 X(28,J) = X(27,J)
1103  NEXT J
1104  FOR I = 0 TO 28
1105 X(I,21) = X(I,20):X(I,0) = X(I,1)
1106  NEXT I
1107  FOR I = A1 TO A2:X(I,0) = 1000
1108  NEXT I: REM OEFFNUNG AUS DER UNTER DRUCK FLUID
    AUSSTROEMT
1111  FOR I = 14 TO 28:X(I,9) = X(I,8):X(I,10) = X(I,
    11): NEXT I: REM HINDERNIS
1118  FOR I = 1 TO 27
1119  FOR J = 1 TO 20
1120  GOSUB 2100: REM KNOTENFUNKTION
1130  NEXT J,I
1132  IF T / 100 <  > INT (T / 100) THEN 1160
1133  POKE 43602,0
1134  PRINT  CHR$ (4)"OPEN DRUCKDATEI."T / 100",S6,D1
    "
1135  PRINT  CHR$ (4)"WRITE DRUCKDATEI."T / 100
1136  FOR I = 0 TO 28: FOR J = 0 TO 21
1137  PRINT X(I,J): NEXT J,I
1138  PRINT  CHR$ (4)"CLOSE DRUCKDATEI."T / 100
1140  PRINT  CHR$ (4)"PR#1"
1141  POKE 1913,66: PRINT "T="T + 16
1142  PRINT  CHR$ (17)
1143  PRINT  CHR$ (4)"PR#0"
1150  REM ----------------------------------------
    ----------------------------
```

```
1160  HGR2
1161  IF H = 0 GOTO 1170
1165  HPLOT 27 * 5 + 9 * 4 - 7,9 * 4 + 90 TO 14 * 5 +
    9 * 4 - 7,9 * 4 + 90 TO 14 * 5 + 10 * 4 - 7,10 *
    4 + 90 TO 27 * 5 + 10 * 4 - 7,10 * 4 + 90: REM HI
    NDERNIS,OPTISCH
1170  FOR I = 1 TO 27
1180  FOR J = 1 TO 20
1190 X(I,J) = XNEU(I,J)
1191 X = I * 5 + J * 4 - 7
1192 Y = J * 4 + 90
1210  HPLOT X,Y TO X,Y - X(I,J) / SK
1220  NEXT J,I
1240  GOTO 1100
1900  END
1990  REM ----------------------------------------
    ----------------------------

1998 :
1999 :
2000  REM --------------------KNOTENFUNKTION---------
    ----------------------------
2010  REM DRUECKE DER BENACHBARTEN VOLUMENELEMENTE WE
    RDEN ARITHMETISCH GEMITTELT
2090  REM ----------------------------------------
    ----------------------------
2100 XNEU(I,J) = (X(I - 1,J) + X(I,J - 1) + X(I,J) +
    X(I,J + 1) + X(I + 1,J)) / N
2120  RETURN
2200  REM ----------------------------------------
    ----------------------------

4998 :
4999 :
5000  REM ----------------RESTART AUS DRUCKDATEI-----
    ----------------------------
5100  PRINT
5110  INPUT "DRUCKDATEI-#? ";T
5133  POKE 43602,0
5134  PRINT  CHR$ (4)"OPEN DRUCKDATEI."T",S6,D1"
5135  PRINT  CHR$ (4)"READ DRUCKDATEI."T
5136  FOR I = 0 TO 28: FOR J = 0 TO 21
5137  INPUT X(I,J): NEXT J,I
5138  PRINT  CHR$ (4)"CLOSE DRUCKDATEI."T
5140 T = T * 100 + 16
5150  GOTO 1100
5200  REM ----------------------------------------
    ----------------------------
```

```
1000  REM ----------------------------------------
    ---------------
1020  REM ------------------WINDSTAERKE-----------
    ----------------------------
1022  REM ----------------------------------------
    ----------------------------
1023  HOME
1024  PRINT "1.GRADIENTENDARSTELLUNG": PRINT "2.DRUCK
    DARSTELLUNG": PRINT
1025  INPUT "DEINE WAHL? ";W
1027  INPUT "ANZAHL DRUCKDATEIEN? ";T1
1028 T1 = T1 - 1
1030  DIM X(30,30)
1085  HCOLOR= 3
1086 N = 1 / 4
1087  FOR T = 0 TO T1
1088  POKE 43602,0
1090  PRINT  CHR$ (4)"OPEN DRUCKDATEI."T",S6,D1"
1100  PRINT  CHR$ (4)"READ DRUCKDATEI."T
1110  FOR I = 0 TO 28: FOR J = 0 TO 21
1120  INPUT X(I,J): NEXT J,I
1130  PRINT  CHR$ (4)"CLOSE DRUCKDATEI."T
1135  ON W GOTO 1147,2100
1140  REM ----------------------------------------
    ---------------
1147  HGR2
1148  HPLOT 270,90 TO 140,90 TO 140,100 TO 270,100: REM
    HINDERNIS
1150  FOR I = 1 TO 27
1160  FOR J = 1 TO 19
1169  REM (X,Y) IST DER VEKTOR DER WINDGESCHWINDIGKEI
    T IM PUNKT (I,J)
1170 X = (X(I - 1,J) - X(I + 1,J)) / 2
1173 Y = (X(I,J - 1) - X(I,J + 1)) / 2
1174 LL =  SQR (X * X + Y * Y)
1180 LL = LL ^ (1 / 5) / (LL + (LL = 0))
1181 X = LL * X / N
1182 Y = LL * Y / N
1183  IF I * 10 + X > 279 THEN X = 279 - 10 * I
1184  IF I * 10 + X < 0 THEN X =  - I * 10
1187  IF J * 10 + Y > 191 THEN Y = 191 - 10 * J
1188  IF J * 10 + Y < 0 THEN Y =  - J * 10
1190  HPLOT I * 10,J * 10 TO I * 10 + X,J * 10 + Y
1195  NEXT J,I
```

```
1201  PRINT  CHR$ (4)"PR#1"
1202  PRINT "T="T * 100 + 16
1203  PRINT  CHR$ (17): REM CTRL-Q
1204  PRINT  CHR$ (4)"PR#0"
1210  NEXT T
1290  GET A$: TEXT
1300  END
1998 :
1999 :
2000  REM ----------------------------------------
    ----------------------------
2100  HGR2
2103  HPLOT 1 * 5 + 1 * 4 - 7,1 * 4 + 90 TO 1 * 5 + 1
    9 * 4 - 7,19 * 4 + 90 TO 27 * 5 + 19 * 4 - 7,19 *
    4 + 90 TO 27 * 5 + 1 * 4 - 7,1 * 4 + 90 TO 1 * 5 +
    1 * 4 - 7,1 * 4 + 90
2105  FOR I = 1 TO 27
2107  HPLOT I * 5 + 4 - 7,4 + 90 - X(I,1) / 15
2110  FOR J = 1 TO 19
2120 X = I * 5 + J * 4 - 7
2130 Y = J * 4 + 90
2140  HPLOT  TO X,Y - X(I,J) / 15
2145  NEXT J,I
2150  FOR J = 1 TO 19
2155  HPLOT 5 + J * 4 - 7,J * 4 + 90 - X(1,J) / 15
2160  FOR I = 1 TO 27
2165 X = I * 5 + J * 4 - 7
2170 Y = J * 4 + 90
2175  HPLOT  TO X,Y - X(I,J) / 15
2180  NEXT I,J
2201  PRINT  CHR$ (4)"PR#1"
2202  PRINT "T="T * 100 + 16
2203  PRINT  CHR$ (17): REM  CTRL-Q
2204  PRINT  CHR$ (4)"PR#0"
2210  NEXT T
2290  GET A$: TEXT
```



T=16





T=116



Schiffsführungs- und Simulationsanlage SUSAN	FACHHOCHSCHULE HAMBURG FACHBEREICH SEEFAHRT	5/85
	Druckausgleich-Modell	S. 6
	Druck und Gradient	



T=216



Schiffsführungs- und Simulationsanlage SUSAN	FACHHOCHSCHULE HAMBURG FACHBEREICH SEEFAHRT	5/85
	Druckausgleich-Modell	S. 7
	Druck und Gradient	



T=316





T=416



Schiffsführungs- und Simulationsanlage SUSAN	FACHHOCHSCHULE HAMBURG FACHBEREICH SEEFAHRT	5/85
	Druckausgleichs-Modell	S. 9
	Druck und Gradient	



T=516



Schiffsführungs- und Simulationsanlage SUSAN	FACHHOCHSCHULE HAMBURG FACHBEREICH SEEFAHRT	5/85
	Druckausgleich-Modell	S.10
	Druck und Gradient	



T=616



Schiffsführungs- und Simulationsanlage SUSAN	FACHHOCHSCHULE HAMBURG FACHBEREICH SEEFAHRT	5/85
	Druckausgleich-Modell	S.11
	Druck und Gradient	



T=716



Schiffsführungs- und Simulationsanlage SUSAN	FACHHOCHSCHULE HAMBURG FACHBEREICH SEEFAHRT	5/85
	Druckausgleich-Modell	S.12
	Druck und Gradient	



T=816





T=916





T=1016